
Abstract

We show that using deductive systems to specify an oine partial evaluator allows its correctness to be
mechanically veri ed.
For a -mix-style partial evaluator, we specify binding-time constraints using a natural-deduction logic,
and the associated program specializer using natural (aka \deductive") semantics. These deductive systems
can be directly encoded in the Elf programming language | a logic programming language based on the
LF logical framework. The speci cations are then executable as logic programs. This provides a prototype
implementation of the partial evaluator.
Moreover, since deductive system proofs are accessible as objects in Elf, many aspects of the partial evaluation correctness proofs (e.g., the correctness of binding-time analysis) can be coded in Elf and mechanically
veri ed.
This work illustrates the utility of declarative programming and of using deductive systems for de ning
program specialization systems: by exploiting the logical character of de nitions, one may specify, prototype,
and mechanically verify correctness via meta-programming | all within a single framework.

1 Introduction
As the application of program-specialization techniques increases, it is important to study ways of formally
specifying and proving these techniques correct. Unfortunately, relatively few specialization systems have been
proven correct, and we are not aware of any that have been mechanically veri ed. Even when systems have
been proven correct, it is unclear from the literature which styles of formalization are preferable.
As an example, consider oine partial evaluation of functional programs [5, 20]. This consists of two
phases: a binding-time analysis phase (where information is gathered about what parts of the source program
depend on known or unknown data), and a specialization phase (where constructs depending on known data are
reduced away). Denotational semantics has been the most widely applied formalism for specifying oine partial
evaluators. The denotational method allows both phases to be speci ed in the same framework: the analysis
phase is speci ed as an abstract interpretation [6, 21] and the specialization phase is speci ed as a function
from a source program to a residual program. Moreover, a prototype implementation can often be obtained
by a fairly straightforward (though informal) transliteration of the denotational speci cations into functional
programs.
Recent work speci es the analysis phase using type systems while the specialization phase is speci ed denotationally [10, 22]. Unfortunately, this approach loses a degree of synergy since both phases are not speci ed in
the same framework. For example, one can no longer obtain a prototype via a simple translation to functional
programs (since the typing rules which specify the analysis cannot be simply translated). Others de ne the

program specializer using operational semantics and thus avoid what Wand has characterized as a \morass of
model-theoretic details" [37] associated with the denotational framework.
Note that the type system and operational semantics formalisms can be uni ed if one emphasizes their logical
character: a type-based analysis is a logic for deducing program properties, and an operational semantics is a
logic for deducing computational steps or input/output behaviour of programs. However, in program specialization systems that use these formalisms, this logical character has neither been emphasized nor exploited.
In this paper, we exploit the logical character of such deductive systems and obtain a uniform framework for
specifying, prototyping, and mechanically verifying the correctness of program specialization systems. Specifically, we consider oine partial evaluation in the style of the partial evaluator -mix [11].1 -mix is a good
illustrative case since it is simple, and one of the few partial evaluators with a rigorous semantic foundation. It
has also spawned additional work on the correctness of binding-time analysis and specialization [27, 37].
The results of the present work are as follows.
We give speci cations of binding-time constraints and specialization as natural-deduction style logics.
These speci cations simplify meta-theory activities such as proving the correctness of binding-time analysis
and specialization.
We formalize the speci cations using LF | a meta-language (a dependently-typed -calculus) for de ning
logics [19]. In LF, judgements (assertions) are represented as types, and deductions are represented as
objects. Determining the validity of a deduction is reduced to checking if the representing object is
well-typed. Since LF type-checking is decidable, purported deductions can be checked automatically for
validity.
We obtain prototypes directly from the formal speci cations using Elf | a logic programming language
based on LF [29]. Elf gives an operational interpretation to LF types by treating them as goals. Thus,
the LF speci cations of the binding-time analysis and specializer are directly executable in Elf.
We formalize and mechanically verify much of the meta-theory of oine partial evaluation (e.g., correctness of binding-time analysis and soundness of the specializer) via meta-programming in Elf. Correctness
conditions are formalized as judgements about \lower-level" deductions describing object-language evaluation and transformation. Proofs of correctness are formalized as deductions of the correctness judgements.
Elf type-checking mechanically veri es that these deductions (and hence the correctness proofs) are valid.2
This methodology of speci cation/implementation/veri cationusing LF and Elf has been successfully applied
in other problem areas [18, 23]. In particular, we build on Hannan and Pfenning's work on compiler veri cation
in Elf [18]. They conjectured that their techniques could also be applied to partial evaluation [18, p. 416]. They
also identify the veri cation of transformations based on ow analyses as a \challenging problem, yet to be
addressed" [18, p. 415]. The present work addresses both of these points. We con rm their conjecture that LF
and Elf can be used for speci cation/implementation/veri cation of partial evaluators. Moreover, we give one
instance where transformations based on ow analyses can be veri ed | namely, the specialization of programs
based on binding-time analysis. Hannan has recently reported on a second instance: the use of Elf to encode a
type system for closure conversion [16].
The rest of the paper is organized as follows. Section 2 summarizes LF and Elf. Section 3 presents the
object language and its encoding in Elf. Section 4 presents the speci cations of the binding-time analysis and
specializer. Section 5 illustrates how these speci cations are executable in Elf. Section 6 shows how metatheoretic properties of the speci cations can be mechanically veri ed. Section 7 surveys related work, and
Section 8 concludes.








2 LF and the Elf Programming Language
This section presents a very brief overview of LF and ELf. The reader is referred to the original work on LF by
Harper, Honsell, Plotkin [19] for a detailed presentation of LF. Pfenning [29] gives the technical foundations of
1 Our setting di ers from that of -mix in two ways: (1) we are not concerned with self-applying the partial evaluator, and (2)
our object language is typed while -mix's is untyped. However, the techniques here apply equally well to the untyped object
language of -mix (in fact, they are simpler in the untyped case).
2 This paper contains all the required Elf code and is implemented using version 0.3 of Elf. The code is available upon request
from the author.

Elf. Papers by Hannan and Pfenning [18] and Michaylov and Pfenning [23] give applications of Elf that provide
the main techniques used in the present work.

2.1 LF | a framework for de ning logics

The LF calculus has three levels: objects, families, and kinds. Families are classi ed by kinds, and objects are
classi ed by types, i.e., families of kind Type.
Kinds
K ::= Type x:A: K
Families A ::= a x:A1: A2 x:A1 : A2 A M
Objects M ::= c x x:A : M M1 M2
Family-level constants are denoted by a, and object-level constants by c. A1 A2 abbreviates x:A1: A2 and
x:A: K when x does not appear free in A2 or K, respectively. The typing rules for LF can be found in [19].
We take -equivalence as the notion of de nitional equality in LF [19, Appendix A.3]. For all the languages
we consider, we identify terms up to renaming of bound variables.
One de nes a logic in LF by specifying a signature which declares the kinds of family-level constants a
and types of object-level constants c. These constants are constructors for the logic's syntax, judgements,
and deductions. Well-formedness is enforced by LF type-checking. The LF type system can represent the
conditions associated with binding operators, with schematic abstraction and instantiation, and with the variable
occurrence and discharge conditions associated with rules in systems of natural deduction.
j

j

j

j

j

j

j

!

2.2 Elf | an implementation of LF

The syntax of Elf is as follows (the last column lists the corresponding LF term). Optional components are
enclosed in .
kindexp ::= type
Type
id :famexp kindexp
x:A: K
famexp -> kindexp
A K
hi

j

f

h

ig

j

!

famexp ::= id
j
fidh:famexp 1 ig famexp 2
j
[idh:famexp1 i] famexp2
j
famexp objexp
j
famexp1 -> famexp2
j
famexp2 <- famexp1
objexp ::= id
j
[idh:famexpi] objexp
j
objexp1 objexp2

a
x:A1: A2
x:A1 : A2
AM
A1 A2
A1 A2
!

!

c
x:A : M
M1 M2
The terminal id ranges over variables, and family and object constants. Bound variables and constants in Elf can
be arbitrary identi ers, but free variables in a declaration or query must begin with an upper case letter. Free
variables act as logic variables and are implicitly -abstracted. Elf's term reconstruction phase (preprocessing)
inserts these abstractions as well as appropriate arguments to these abstractions. It also lls in the omitted
types in quanti cations x and abstractions [x] and omitted types or objects indicated by an underscore _.
The <- is used to improve the readability of some Elf programs. B <- A is parsed into the same representation
as A -> B; -> is right associative, while <- is left associative. An Elf program is a representation of an LF
signature. Although we are implicitly encoding logics in LF, we give all encodings using the syntax of Elf.
f g

3 The Object Language
3.1 Syntax

3

Object language:

e
Exp
e ::= x 0 sc e pr e if0 e1 e2 e3
lam x : e app e0 e1 x x: e
2

j

Elf encoding:

j

j

j

j


Typ
 ::= nat 1
2

j

j

!

2

j

exp : type.

typ : type.

z : exp.
sc : exp -> exp.
pr : exp -> exp.
if : exp -> exp -> exp -> exp.
lam : (exp -> exp) -> exp.
app : exp -> exp -> exp.
fix : (exp -> exp) -> exp.

nat : typ.
arrow : typ -> typ -> typ.

Figure 1: The object language 
Figure 1 presents the syntax of the object language  (a PCF-like language) and its encoding in Elf. We
identify object-level terms up to -equivalence (i.e., up to renaming of bound variables). We write e1 e2
when e1 and e2 are -equivalent.
The Elf signature for the object language syntax includes family constants exp and typ, and object constants
for each term and type constructor. Variable binding in lam x : e and x x: e is represented using binding in the
meta-language (i.e., using higher-order abstract syntax [31]) | an idea going back to Church [3]. For example,
the expression
app (lam x1 : lam x2 : app (lam x3 : x3) x1) 0
is encoded as
app (lam [x1] lam [x2] app (lam [x3] x3) x1) z:
In this style of encoding, the body of the lam expression


lam x1 : lam x2 : app (lam x3 : x3) x1

is represented as a \higher-order term" that maps expressions to expressions. This allows substitution in the
object language to be represented using Elf application and -reduction. For example,
(lam x2 : app (lam x3 : x3) x1 )[x1 := z]
is represented in Elf as

([x1] lam [x2] app (lam [x3] x3) x1) z:
An Elf -reduction gives the expected result of the substitution.
lam [x2] app (lam [x3] x3) z
This representation properly implements capture-free object-level substitution since it is expressed via Elf
substitution (which the Elf implementation guarantees to be capture-free). Note that since variables in the
object language are identi ed with variables in the meta-language, there is no explicit representation of identi ers
in the Elf signature for the object language.
Technically, one must ensure that encodings of syntax into Elf are adequate, i.e., that there is a compositional
bijection between the syntactic entities in the logical system and well-formed LF long -normal forms under the
given signature. An adequate encoding ensures that each entity is encoded uniquely and that no representations
of additional entities are introduced. All the encodings we use are adequate. See Harper et al.[19] for a detailed
discussion and proofs of adequacy for encodings similar to the ones used here.

tp z :

tp 0 : nat

tp sc :

tp e : nat
tp sc e : nat

tp pr :

tp e : nat
tp pr e : nat

tp cnd :

tp e1 : nat tp e2 :  tp e3 : 
tp if0 e1 e2 e3 : 

tp lam :

(tp x : 1 )
tp e : 2
tp lam x : e : 1 2
!

tp app :

tp e0 : 1 ! 2 tp e1 : 1 w1'1
tp app e0 e1 : 2

tp x :

(tp x : )
tp e : 
tp x x: e : 

Figure 2: The typing rules for 

3.2 Typing
3.2.1 Natural deduction style rules

Figure 3 presents the typing rules for  in a natural-deduction style logic (we use the notation of Prawitz
[33]).3 Parentheses in the hypotheses of the rules tp lam and tp x indicate the discharging of zero or more
occurrences of assumptions. We write T tp e :  when tp e :  is derivable under undischarged assumptions
T. We shall only be concerned with assumptions involving identi ers (e.g., tp x : ) since the logic only
allows discharging of assumptions of this form. For simplicity, we require identi ers xi in assumption sets
T = tp x1 : 1 ; :::; tp xn : n , to be pairwise distinct.
`

f

g

3.2.2 Elf encoding

Figure 3 presents the Elf encoding of the typing rules. The gure begins with a declaration of the typing
judgement tp as a relation between source expressions exp and types typ. The implicit universal quanti cation
of the upper case variables captures the schematic nature of the rules. The rules for binding constructs lam
and fix (which involve hypothetical proofs in the premises) are encoded in Elf as proof constructors that take
proofs of hypothetical judgements as arguments. Such proofs are represented as functions mapping proofs of
assumption judgements to proofs of consequent judgements. The higher-order syntax representation requires
that judgements involving identi ers be represented as schematic judgements (i.e., identi ers are -quanti ed).
This is the case in rules tp lam and tp fix where e.g., the judgement tp (E x) T2 expresses that E may be
instantiated to representations of terms with free occurrences of x [19, Section 3.1].
3

See the textbook of Tennent [36] for extensive examples using natural deduction logics to specify syntax and typing.

%%% typing judgement
tp : exp -> typ -> type.
%%% typing axioms and rules
tp_z : tp z nat.
tp_sc : tp (sc E) nat
<- tp E nat.
tp_pr : tp (pr E) nat
<- tp E nat.
tp_if : tp
<<<-

(if E1 E2 E3) T
tp E1 nat
tp E2 T
tp E3 T.

tp_lam : tp (lam E) (arrow T1 T2)
<- {x:exp} tp x T1 -> tp (E x) T2.
tp_app : tp (app E0 E1) T2
<- tp E0 (arrow T1 T2)
<- tp E1 T1.
tp_fix : tp (fix E) (arrow T1 T2)
<- {x:exp} tp x (arrow T1 T2) -> tp (E x) (arrow T1 T2).

Figure 3: The Elf encoding of the typing rules

int z :

0 +int 0

int sc :

e int v
+

sc e +int sc v

int prz :

e +int 0
pr e +int 0

int prsc :

e +int sc v
pr e +int v

int ifz :

e1 int 0 e2 int v
if0 e1 e2 e3 int v
+

+

+

int ifsc :

e1 int sc v1 e3 int v
if0 e1 e2 e3 int v
+

+

+

int lam :
int app :

lam y : e +int lam y : e

e0 int lam y : e00 e00 [y := e1 ] int v
app e0 e1 int v
+

+

+

int x :

e[y := x y: e] int v
x y: e int v
+

+

Figure 4: The natural semantics rules for  (interpretation logic)

3.2.3 Executing the speci cation

The Elf encoding of the typing rules (see Figure 3) is an executable logic program. For example, type inference
for the example expression of Section 3.1 can be performed via the following Elf query.
?- tp (app (lam [x1] lam [x2] app (lam [x3] x3) x1) z) T.
Solving...
T = arrow T1 nat.
;
no more solutions

The solution T = arrow T1
for any argument type T1.

nat

nat

indicates that the example expression is a function yielding results of type

3.3 Operational Semantics
3.3.1 Natural deduction style rules

Figure 4 presents the call-by-name natural semantics rules (i.e., the interpretation or evaluation logic) for 
terms. These rules correspond to the usual call-by-name \natural" or \deductive" operational semantics for
PCF [14, Chapter 4]. We write e int a when e int a is derivable, and e int when there does not exist an
a such that e int a. It is easy to check that the relation induced by int is a partial function.
Two terms are considered operationally equivalent if they cannot be distinguished by any program context
[14, Chapter 6].
`

`

+

+

*

+

+

%%% interpreter judgement
int : exp -> exp -> type.
%%% interpreter axioms and rules
int_z : int z z.
int_sc : int (sc E) (sc A)
<- int E A.
int_prz : int (pr E) z
<- int E z.
int_prsc : int (pr E) A
<- int E (sc A).
int_ifz : int (if E1 E2 E3) A
<- int E1 z
<- int E2 A.
int_ifsc : int (if E1 E2 E3) A
<- int E1 (sc A1)
<- int E3 A.
int_lam : int (lam E) (lam E).
int_app : int (app E0 E1) A
<- int E0 (lam E0')
<- int (E0' E1) A.
int_fix : int (fix E) A
<- int (E (fix E)) A.

Figure 5: The Elf encoding of the natural semantics rules (interpreter)

De nition 1 (Operational Equivalence) e1 ; e2  are operationally equivalent (written e1
2

contexts C such that ` tp e1 : nat and ` tp e2 : nat one of the following holds:
1. e1 *int and e2 *int , or
2. ` e1 +int a1 and ` e2 +int a2 and a1  a2.



e2 if for all

Section 6 shows that our partial evaluator produces an output program that is operationally equivalent to the
given input program.

3.3.2 Elf encoding

Figure 5 gives the Elf encoding of the natural semantics. The encoding techniques are similar to those used by
Michaylov and Pfenning [23]. In the rules int app and int fix, we take advantage of the higher-order abstract
syntax representation and use Elf application (i.e., -reduction) to implement capture-free substitution.

Value rules:

Elf encoding:

val z :

val 0

val sc :

val v
val sc v

%%% value judgement
val : exp -> type.
%%% value logic

val lam : val lam x : e

val_z : val z.
val_sc : val (sc E)
<- val E.
val_lam : val (lam E).

Figure 6: The value rules and Elf encoding

3.3.3 Executing the speci cation

As with the typing rules, the Elf encoding of the natural semantics rules (see Figure 5) is an executable logic
program. For example, evaluation of the example expression of Section 3.1 can be performed via the following
Elf query.
?- int (app (lam [x1] lam [x2] app (lam [x3] x3) x1) z) A.
A = lam [x2:exp] app (lam [x3:exp] x3) z.
;
no more solutions

3.3.4 Canonical terms of evaluation

An examination of the rules of Figure 4 reveals that the canonical terms (i.e., values) of evaluation are either
0, sc e where e is also canonical, and lam x : e0. We formalize the notion of canonical term with the judgement
val of Figure 6. The encoding of the value rules in Elf is straightforward.
The following proposition formalizes the above claim: terms satisfying the value judgement val are indeed

the results of evaluation.

Proposition 1 If e
`

+

int

e0 , then

`

val e0

The proof of this proposition proceeds by induction over derivations of e int e0 ; it can be formalized and
mechanically checked in Elf. The reader is referred to the work of Michaylov and Pfenning [23] where a detailed
presentation of such a proof for call-by-value evaluation is given.
Finally, note that we have followed convention [14] and de ned the value judgement on the syntactic category
of expressions while omitting any notion of typing. Thus, the judgement does not enforce well-typedness and
admits terms that would not result from the evaluation of a well-typed program. For example, val sc (lam x : x)
but there does not exist a  such that tp sc (lam x : x) : . Well-typedness of values is expressed by using the
typing judgement in conjunction with the value judgement.
+

`

`

4 Specifying an Oine Partial Evaluator for 
A partial evaluator takes a source program p and a subset s of p's input, and produces a residual program ps
which is specialized with respect to s. The correctness of the partial evaluator implies that running ps on p's
remaining input d gives the same result as running p on the complete input s and d. The data s and d are
9

Annotated object language:
w 2 Sexp
w ::= y j 0m j scm w j prm w j if0m w1 w2 w3
lamm y : w j appm w0 w1 j xm y: w j
lift w
Elf encoding:

m
Sder
m ::= s d
2

j

j

sexp : type.

sder : type.

bz : sder -> sexp.
bsc : sder -> sexp -> sexp.
bpr : sder -> sexp -> sexp.
bif : sder -> sexp -> sexp -> sexp -> sexp.
blam : sder -> (sexp -> sexp) -> sexp.
bapp : sder -> sexp -> sexp -> sexp.
bfix : sder -> (sexp -> sexp) -> sexp.
lift : sexp -> sexp.

s : sder.
d : sder.

Figure 7: The annotated object language bt
often referred to as static and dynamic data (respectively) since s is xed at specialization time whereas one
may supply various data d during runs of ps .
The specialized program ps is obtained from p by evaluating constructs that depend only on s, while rebuilding constructs that may depend on dynamic data. Oine partial evaluation accomplishes this in two phases:
(1) a binding-time analysis phase, and (2) a specialization phase.
1. Binding-time analysis: Given assumptions about which program inputs are static and dynamic,
binding-time analysis constructs an annotated program where each program construct is annotated with
a specialization directive and a specialization type.




Specialization directives: A construct is assigned a directive of eliminable if it depends only on

static data and thus can be completely evaluated during the specialization phase. A construct is
assigned a directive of residual if it may depend on dynamic data and thus must be reconstructed in
the specialization phase.
Specialization types: The specialization types (a.k.a binding times) are the carriers of information
during the analysis phase. They describe the \knownness" or the \unknownness" of expressions. This
information is used to determine the specialization directives assigned to constructs. For example, if
the argument of a destructor construct has a specialization type indicating that it is unknown, then
that construct cannot be evaluated at specialization time and must be given a residual directive.

2. Specialization: During the specialization phase, the specializer simply follows the directives assigned during binding-time analysis: eliminable constructs are evaluated (and thus eliminated); residual constructs
are reconstructed (and thus appear in the residual program).
Several di erent formalizations of binding-time analysis and specialization are given in the literature (see
the textbook of Jones et al. [20] for a survey). In the remainder of this section, we give a logical formalization
that is suitable for encoding in Elf.

4.1 Binding-time analysis

We rst outline how binding-time information is expressed in a program annotated with specialization directives and types. Following this, we give a binding-time logic that determines which annotations are appropriate
for a given soure language term.
10

Specialization types:

Elf encoding:

'nat
Styp[nat]
'nat ::= sta dynnat

styp

2

j

Styp[1 2]
'1 !2
'1 !2 ::= '1 '2 dyn1 !2
2

!

!

j

: typ -> type.

sta
: styp nat.
dyn
: styp T.
barrow : styp T1 -> styp T2 -> styp (arrow T1 T2).

Figure 8: The specialization types for bt

4.1.1 Specialization directives

A binding-time analysis for  associates each  term with a term in the annotated language bt of Figure 7. An
annotated term is indexed by a specialization directive s or d indicating if it is eliminable (i.e., it depends only
on static data and thus can be eliminated at specialization time) or residual (i.e., it may depend on dynamic
data and thus must be placed in the residual program). Identi ers are not indexed since the appropriate
information can be determined from the environment. A coercion construct lift is added to bt to residualize
the result of evaluating an eliminable term. This allows static computation to occur in a residual context (an
example is given in Section 4.3). A term w bt is completely residual if it consists of only d-indexed constructs
and identi ers. For example, the following well-annotated version of the example expression of Section 3.1 is
completely residual.
appd (lamd y1 : lamd y2 : appd (lamd y3 : y3) y1 ) 0d
Intuitively, the specializer will output completely residual terms | all eliminable constructs will have been
evaluated (this will be proven in Section 6.1). The Elf encoding of bt terms follows that of  terms (except
that directive indexing is captured by supplying an extra argument of type sder to a constructor). For example,
the term above is encoded as follows.
2

bapp d (blam d [y1] (blam d [y2] (bapp d (blam d [y3] y3) y1))) (bz d)

4.1.2 Specialization types

Figure 8 presents a -indexed family of specialization types for bt (we omit type indices on specialization types
when they can be inferred from the context). A specialization type ' is dynamic if ' = dyn; otherwise ' is
static. The intention is that
sta Styp[nat] will tag expressions of type nat that are guaranteed to evaluate to known data (.e., numerals
0, sc 0, etc.);
dynnat Styp[nat] will tag expressions of type nat whose evaluation may depend on unknown data and
thus cannot be guaranteed to evaluate to numerals;
'1 '2 Styp[1 2] will tag expressions of type 1 2 that are guaranteed to evaluate to known
data (.e., abstractions);
dyn1 !2 Styp[1 2 ] will tag expressions of type 1 2 whose evaluation may depend on unknown
data and thus cannot be guaranteed to evaluate to an abstraction.
Specialization types are encoded in Elf via the type family styp:typ -> type. For the dyn and barrow constructors, the type indices are encoded as logic variables T, T1, and T2, which will be instantiated by uni cation.
This formalizes the above convention of allowing type indices to be inferred from the context.


2







2

!

2

2

!

!

!

!

4.1.3 Binding-time logic

Figure 9 presents a natural-deduction-style logic for deriving judgements of the form bta e :  [w : ' ]. Parentheses in the hypotheses of the rules for binding constructs (i.e., bta lam s, bta lam d, bta x s, bta x d) indicate
11

the discharging of zero or more occurrences of assumptions. We write ? bta e :  [w : ' ] when bta e :  [w : ' ]
is derivable under undischarged assumptions ?.
Derivable judgements specify constraints that an actual binding-time analysis algorithm must satisfy. Intuitively, if ? bta e :  [w : ' ], then given initial binding-time assumptions ?, a binding-time analysis algorithm
may map e Terms[] of type  Types[] to a directive annotated term w Terms[bt] of specialization type
' Spec-types[]. Specifying the analysis in this way allows one to reason about correctness of the analysis
independently of the actual algorithm | proving correctness with respect to the constraints implies that any
algorithm satisfying the constraints will be correct. Furthermore, Section 5.1 shows how the Elf proof search
mechanism can be used to extract algorithmic content directly from the logic specifying the constraints.
We only consider assumptions involving identi ers (e.g., bta x :  [y : ' ]) since the logic only allows discharging of assumptions of this form. For ? = bta x1 : 1 [y1 : '1 ]; :::; bta xn : n [yn : 'n ] , the xi are
required to be pairwise distinct (similarly for the yi ). A static assumption (resp. dynamic assumption) is an
assumption bta x :  [y : ' ] where ' is static (resp. dynamic). ?s denotes a set of only static assumptions; ?d
denotes a set of only dynamic assumptions.
A simple induction over the structure of deductions for ? bta e :  [w : ' ] shows that the relation between
e and w is one-to-many, i.e., there may be many valid annotations of e. As a consequence, a binding-time
analysis satisfying these constraints has some exibility in assigning correct annotations. In practice, one
desires an analysis that gives as many eliminable directives (i.e., gives as many s indices) as possible.4 The
constraint system is complete in the sense that one may always obtain a valid annotation of a type correct e by
annotating all components as residual.
Finally, since the relation induced by bta is one-to-many from  to bt, it is many-to-one (i.e., a function)
from bt to . This is essentially an annotation forgetting function that removes directives and lift constructs.
The binding-time judgement is encoded in Elf as follows.
`

`

2

2

2

2

g

f

`

:

bta : exp -> ft typg sexp -> styp t -> type.

The dependent function type (i.e., t : typ :::) expresses the dependency of the indexed specialization type
on the type of the  expression.
Figure 10 presents the Elf encoding of the binding-time logic. The implicit universal quanti cation of the
upper case variables captures the schematic nature of the rules. The rules for binding constructs (which involve
hypothetical proofs in the premises) are encoded in Elf as proof constructors that take proofs of hypothetical
judgements as arguments. Such proofs are represented as functions mapping proofs of assumption judgements
to proofs of consequent judgements. The higher-order syntax representations requires that judgements involving
identi ers be represented as schematic judgements (i.e., identi ers are -quanti ed). This is the case in rules
bta lam s, bta lam d, bta fix s, and bta fix d where e.g., the judgement bta (E x) T2 (W y) P2 expresses
that E and W may be instantiated to representations of terms with free occurrences of x and y respectively [19,
Section 3.1].
f

g

4.2 Specialization
4.2.1 Natural deduction rules

Figures 11 and 12 present the specialization logic for bt terms.
Figure 11 describes the evaluation of eliminable constructs. These rules correspond to the usual call-byname \natural" or \deductive" operational semantics for  given in Figure 4.
Figure 12 describes the reconstruction of residual constructs after subexpressions have been specialized.
The rules for lamd and xd are noteworthy because they introduce operations on open terms (e.g., bodies
of lamd and xd constructs). Following Hannan [15, p. 144], we specialize the body of a lamd (and xd )
under the assumption that the bound variable evaluates to itself.




4 We omit considering whether a term has a \best" annotation according to the logic. See the work on -mix [10, 11] for related
discussions.

bta z s :

bta 0 : nat [0s : sta]

bta sc s :

bta e : nat [w : sta]
bta sc e : nat [scs w : sta]

bta pr s :

bta e : nat [w : sta]
bta pr e : nat [prs w : sta]

bta if s :

bta e1 : nat [w1 : sta] bta e2 :  [w2 : '] bta e3 :  [w3 : ']
bta if0 e1 e2 e3 :  [if0s w1 w2 w3 : ']

bta lam s :

(bta x : 1 [y : '1 ])
bta e : 2 [w : '2 ]
bta lam x : e : 1 2 [lams y : w : '1
!

!

'2 ]

bta app s :

bta e0 : 1 ! 2 [w0 : '1 ! '2 ] bta e1 : 1 [w1 : '1 ]
bta app e0 e1 : 2 [apps w0 w1 : '2 ]

bta x s :

(bta x :  [y : '])
bta e :  [w : ']
bta x x: e :  [ xs y: w : ']

bta z d :

bta 0 : nat [0d : dyn]

bta sc d :

bta e : nat [w : dyn]
bta sc e : nat [scd w : dyn]

bta pr d :

bta e : nat [w : dyn]
bta pr e : nat [prd w : dyn]

bta if d :

bta e1 : nat [w1 : dyn] bta e2 :  [w2 : dyn] bta e3 :  [w3 : dyn]
bta if0 e1 e2 e3 :  [if0d w1 w2 w3 : dyn]

bta lam d :

(bta x : 1 [y : dyn])
bta e : 2 [w : dyn]
bta lam x : e : 1 2 [lamd y : w : dyn]
!

bta app d :

bta e0 : 1 ! 2 [w0 : dyn] bta e1 : 1 [w1 : dyn]
bta app e0 e1 : 2 [appd w0 w1 : dyn]

bta x d :

(bta x :  [y : dyn])
bta e :  [w : dyn]
bta x x: e :  [ xd y: w : dyn]

bta lift :

bta e : nat [w : sta]
bta e : nat [lift w : dyn]

Figure 9: The binding-time logic

bta_z_s : bta z nat (bz s) sta.
bta_sc_s : bta (sc E) nat (bsc s W) sta
<- bta E nat W sta.
bta_pr_s : bta (pr E) nat (bpr s W) sta
<- bta E nat W sta.
bta_if_s : bta (if E1 E2 E3) T (bif s W1 W2 W3) P
<- bta E1 nat W1 sta
<- bta E2 T W2 P
<- bta E3 T W3 P.
bta_lam_s : bta (lam E) (arrow T1 T2) (blam s W) (barrow P1 P2)
<- {x} {y} bta x T1 y P1 -> bta (E x) T2 (W y) P2.
bta_app_s : bta (app E0 E1) T2 (bapp s W0 W1) P2
<- bta E0 (arrow T1 T2) W0 (barrow P1 P2)
<- bta E1 T1 W1 P1.
bta_fix_s : bta (fix E) (arrow T1 T2) (bfix s W) (barrow P1 P2)
<- {x} {y} (bta x (arrow T1 T2) y (barrow P1 P2))
-> (bta (E x) (arrow T1 T2) (W y) (barrow P1 P2)).
bta_z_d : bta z nat (bz d) dyn.
bta_sc_d : bta (sc E) nat (bsc d W) dyn
<- bta E nat W dyn.
bta_pr_d : bta (pr E) nat (bpr d W) dyn
<- bta E nat W dyn.
bta_if_d : bta (if E1 E2 E3) T (bif d W1 W2 W3) dyn
<- bta E1 nat W1 dyn
<- bta E2 T W2 dyn
<- bta E3 T W3 dyn.
bta_lam_d : bta (lam E) (arrow T1 T2) (blam d W) dyn
<- {x} {y} (bta x T1 y dyn) -> (bta (E x) T2 (W y) dyn).
bta_app_d : bta (app E0 E1) T2 (bapp d W0 W1) dyn
<- bta E0 (arrow T1 T2) W0 dyn
<- bta E1 T1 W1 dyn.
bta_fix_d : bta (fix E) (arrow T1 T2) (bfix d W) dyn
<- {x} {y} (bta x (arrow T1 T2) y dyn)
-> (bta (E x) (arrow T1 T2) (W y) dyn).
bta_lift : bta E nat (lift W) dyn
<- bta E nat W sta.

Figure 10: The Elf encoding of the binding-time logic

spec z s :

0s +spec 0s

spec sc s :

w spec a
+

scs w +spec scs a

spec prz s :

w +spec 0s
prs w +spec 0s

spec prsc s :

w +spec scs a
prs w +spec a

spec ifz s :

w1 spec 0s w2 spec a
if0s w1 w2 w3 spec a
+

+

+

spec ifsc s :

w1 spec scs a1 w3 spec a
if0s w1 w2 w3 spec a
+

+

+

spec lam s :
spec app s :

lams y : w +spec lams y : w

w0 spec lams y : w00 w00 [y := w1 ] spec a
apps w0 w1 spec a
+

+

+

spec x s :

w[y := xs y: w] spec a
xs y: w spec a
+

+

Figure 11: The specialization logic (static constructs)

spec z d :

0d +spec 0d

spec sc d :

w spec a
+

scd w +spec scd a

spec pr d :
spec if d :

w spec a
+

prd w +spec prd a

w1 spec a1 w2 spec a2 w3 spec a3
if0d w1 w2 w3 spec if0d a1 a2 a3
+

+

+

+

spec lam d :

(y spec y)
w spec a
lamd y : w spec lamd y : a
+

+

+

spec app d :

w0 spec a0 w1 spec a1
appd w0 w1 spec appd a0 a1
+

+

+

(y spec y)
w spec a
xd y: w spec xd y: a

spec x d :

+

+

+

spec lift :
spec lift sc :

w +spec 0s
lift w +spec 0d

w spec scs a0 lift a0 spec a
lift w spec scd a
+

+

+

Figure 12: The specialization logic (dynamic and coercion constructs)

spec_z_s : spec (bz s) (bz s).
spec_sc_s : spec (bsc s W) (bsc s A)
<- spec W A.
spec_prz_s : spec (bpr s W) (bz s)
<- spec W (bz s).
spec_prsc_s : spec (bpr s W) A
<- spec W (bsc s A).
spec_ifz_s : spec (bif s W1 W2 W3) A
<- spec W1 (bz s)
<- spec W2 A.
spec_ifsc_s : spec (bif s W1 W2 W3) A
<- spec W1 (bsc s A1)
<- spec W3 A.
spec_lam_s : spec (blam s W) (blam s W).
spec_app_s : spec (bapp s W0 W1) A
<- spec W0 (blam s W0')
<- spec (W0' W1) A.
spec_fix_s : spec (bfix s W) A
<- spec (W (bfix s W)) A.

Figure 13: The Elf encoding of the specialization logic (static constructs)

spec_z_d : spec (bz d) (bz d).
spec_sc_d : spec (bsc d W) (bsc d A)
<- spec W A.
spec_pr_d : spec (bpr d W) (bpr d A)
<- spec W A.
spec_if_d : spec (bif d W1 W2 W3) (bif d A1 A2 A3)
<- spec W1 A1
<- spec W2 A2
<- spec W3 A3.
spec_lam_d : spec (blam d W) (blam d A)
<- {x} spec x x -> spec (W x) (A x).
spec_app_d : spec (bapp d W0 W1) (bapp d A0 A1)
<- spec W0 A0
<- spec W1 A1.
spec_fix_d : spec (bfix d W) (bfix d A)
<- {x} spec x x -> spec (W x) (A x).
spec_lift : spec (lift W) (bz d)
<- spec W (bz s).
spec_liftsc : spec (lift W) (bsc d A)
<- spec W (bsc s A')
<- spec (lift A') A.

Figure 14: The Elf encoding of the specialization logic (dynamic and coercion constructs)

An alternative to introducing local assumptions for bound variables would be to add the axiom y spec
y (which states that all variables evaluate to themselves). However, this approach cannot be directly
represented in Elf. Under the higher-order abstract syntax representation, the variables y are not objects
in the Elf signature. The approach taken in Figure 12 seems the most natural with the higher-order
abstract syntax representation.
The nal rules of Figure 12 coerce an eliminable numeral to a residual expression.
We write  w spec a when w spec a is derivable under undischarged assumptions . Intuitively, if
 w spec a, then the specializer maps w Terms[bt] to answer a Terms[bt] in context . We only
consider assumptions involving identi ers (e.g., y spec y) since the logic only allows discharging of assumptions
of this form. For  = y1 spec y1 ; :::; yn spec yn the yi are required to be pairwise distinct.  is compatible
with ? = bta x1 : 1 [y1 : '1 ]; :::; bta xn : n [yn : 'n ] if  = y1 spec y1 ; :::; yn spec yn . It is easy to
check that the relation induced by spec is a partial function (given the constraints on assumptions above).
+

+

4.2.2 Elf encoding

Figures 13 and 14 give the Elf encoding of the specialization logic. The encoding techniques are similar to those
used in the previous section. The specialization judgement speci es a binary relation (which is actually a partial
function) on annotated terms Sexp and is encoded in Elf as follows.
spec : sexp -> sexp -> type.

In the rules spec app s and spec x s, we take advantage of the higher-order abstract syntax representation
and use Elf application (i.e., -reduction) to implement capture-free substitution.

4.2.3 Canonical terms of specialization

In the conventional deductive semantics for  in Section 3.3, one has numerals (e.g., 0, sc 0) and lam x : e as
canonical terms, i.e., these terms are the results of evaluation. This was formalized by the val judgement.
Intuitively, a specializer is part evaluator and part compiler. Therefore, the canonical terms of the specializer are numerals and lams y : w (corresponding to evaluation results), and completely residual terms or code
(corresponding to compilation results). In a manner analogous to the formalization of values (canonical terms
of evaluation), Figure 15 formalizes the notion of answers (canonical terms of specialization) by de ning a
judgement ans w : ' .
We write A ans w : ' when ans w : ' is derivable under undischarged assumptions A. A is compatible
with ? = bta x1 : 1 [y1 : '1 ]; :::; bta xn : n [yn : 'n ] if A = ans y1 : '1 ; :::; ans yn : 'n . Intuitively,
if ans w : ' holds, then w is a canonical term of specialization type ' . In particular, if ans w : dyn, then
w is completely residual.
In contrast with the val judgement of Section 3.3 which contained no notion of typing, the judgement
ans w : ' relates a canonical term w to a particular specialization type ' . We will need this additional
information when stating the correctness of binding-time analysis in Section 6.1. As with the val judgement,
the ans judgement admits answers that are not well-annotated. For example, ans sc (lam x : x) : dyn but
there does not exist e and  such that bta e :  [sc (lam x : x) : ]. Well-annotatedness of answers is expressed
by using the binding-time judgement in conjunction with the answer judgement. Figure 16 gives the Elf encoding
of the answer logic.
In Section 3.3, Proposition 1 showed that the value judgement val described the results of evaluation. An
analogous proposition below shows that the answer judgement describes the results of specialization.
`

g

f

g

f

`

`

`

Proposition 2 If w
`

+

spec

a and

`

bta e :  [w : ' ], then

`

ans a : ' .

The proof of this proposition follows as a corrollary of the mechanically veri ed proof of Theorem 1 in
Section 6.

ans z s :

ans 0s : sta

ans sc s :

ans w : sta
ans scs w : sta

ans lam s :

ans lams x : w : '1 ! '2

ans z d :

ans 0d : dyn

ans sc d :

ans w : dyn
ans scd w : dyn

ans pr d :

ans w : dyn
ans prd w : dyn

ans if d :

ans w1 : dyn ans w2 : dyn ans w3 : dyn
ans if0d w1 w2 w3 : dyn

ans lam d :

(ans x : dyn)
ans w : dyn
ans lamd x : w : dyn

ans app d :

ans w0 : dyn ans w1 : dyn
ans appd w0 w1 : dyn

ans x d :

(ans x : dyn)
ans w : dyn
ans xd x: w : dyn
Figure 15: The answer logic

%%% answer judgement
ans : sexp -> styp T -> type.
%%% answer logic
% static answers
ans_z_s : ans (bz s) sta.
ans_sc_s : ans (bsc s W) sta
<- ans W sta.
ans_lam_s : ans (blam s W) (barrow P1 P2).
% dynamic answers
ans_z_d : ans (bz d) dyn.
ans_sc_d : ans (bsc d W) dyn
<- ans W dyn.
ans_pr_d : ans (bpr d W) dyn
<- ans W dyn.
ans_if_d : ans (bif d W1 W2 W3) dyn
<- ans W1 dyn
<- ans W2 dyn
<- ans W3 dyn.
ans_lam_d : ans (blam d W) dyn
<- {x} ans x dyn -> ans (W x) dyn.
ans_app_d : ans (bapp d W0 W1) dyn
<- ans W0 dyn
<- ans W1 dyn.
ans_fix_d : ans (bfix d W) dyn
<- {x} ans x dyn -> ans (W x) dyn.

Figure 16: The Elf encoding of the answer logic

4.3 Partial evaluation

We outline how the logics above de ne oine partial evaluation using the following object term.
e def
= app (lam x1 : app x2 (app (lam x3 : x3) x1)) x0
The free variables x0, x2 represent input parameters. The assumptions ?s = bta x0 : nat [y0 : sta] and ?d =
bta x2 : nat nat [y2 : dyn] identify x0 of type nat as known and x2 of type nat nat as unknown. They also
associate x0 and x2 with annotated language identi ers y0 and y2 . The fact that ?s ?d bta e : nat [w : dyn]
where
w def
= apps (lams y1 : appd y2 (lift (apps (lams y3 : y3 ) y1 ))) y0
expresses that a binding-time analysis may associate e with w based on assumptions ?s ?d .
To prepare for specialization, we supply known input via substitution.
f

f

!

g

g

!

[

`

[

?d bta e[x0 := 0] : nat [w[y0 := 0s ] : dyn]
`

Now taking  = y2 spec y2 compatible with ?d (expressing that the dynamic parameter evaluates to itself),
the specialization logic gives
 w[y0 := 0s ] spec appd y2 0d :
Theorem 1 (correctness of binding-time analyis) of Section 6.1 tells us there exists e0
 such that
?d bta e0 : nat [appd y2 0d : dyn]. As noted in Section 4.1, e0 must be app x2 0 | the unannotated version
of appd y2 0d . Theorem 3 (soundness of specialization) of Section 6.2 tells us that e[x0 := 0] is convertible to e0
(denoted e[x0 := 0] = e0 ) in the program calculus for . Thus, for all closed inputs d  of type nat nat,
f

+

g

`

+

2

`

2

!

e[x0 := 0 ; x2 := d] = e0 [x2 := d]:

This re ects the correctness criteria for partial evaluation given at the beginning of Section 4: running e on
static input 0 and dynamic input d is operationally equivalent to running e0 on d.

5 Prototyping an Oine Partial Evaluator for 
5.1 Prototyping a binding-time analysis

The Elf encoding of the binding-time logic (see Figure 10) gives a prototype of the binding-time analysis. The
following Elf query returns all the possible annotations that may be assigned to the example term of Section 3
(here we make dyn the specialization type of the entire term).5
?- bta (app (lam [x1] lam [x2] app (lam [x3] x3) x1) z) (arrow nat nat) W dyn.
Solving...
W = bapp s
(blam s ([y1:sexp] blam d ([y2:sexp] bapp s (blam s ([y3:sexp] y3)) (lift y1))))
(bz s).
W = bapp s
(blam s ([y1:sexp] blam d ([y2:sexp] lift (bapp s (blam s ([y3:sexp] y3)) y1))))
(bz s).

There are actually twelve correct annotations; we show only the two above.

5.2 Prototyping a specializer

The Elf encoding of the specialization logic (see Figures 13 and 14) gives a prototype of the specializer. The
following Elf query returns the result A of specializing the second annotated term above (there is only one answer
since spec is a function).
5

Some of the results of Elf evaluation are -converted for clarity.

?- S : spec (bapp s
(blam s ([y1:sexp] blam d ([y2:sexp] lift (bapp s (blam s ([y3:sexp] y3)) y1))))
(bz s)) A.
Solving...
A = blam d ([y2:sexp] bz d),
S = spec_app_s
(spec_lam_d [y2:sexp] [S:spec y2 y2] spec_lift (spec_app_s spec_z_s spec_lam_s))
spec_lam_s.
;
no more solutions

In this query, we add the variable S which binds to the specialization deduction used to obtain A.6

6 Verifying an Oine Partial Evaluator for 
6.1 Binding-time analysis

A binding-time analysis is correct if it always produces consistent specialization directives. Directives are
consistent if the specializer can never \go wrong". As described by Palsberg [27], a specializer may go wrong
for two reasons: 1) it trusts a part of the program to be eliminable when in fact it is residual, and 2) it trusts
a part of the program to be residual when in fact it is eliminable.
The specializer goes wrong for the rst reason on apps (lamd x : x) 0s . In this case, it attempts evaluation using
the rule spec app s (see Figure 11) but \hangs" (i.e., the result of specialization is unde ned) since lamd x : x is
not an eliminable abstraction. Similar problems may occur with the rule spec lift (e.g., if w spec 0d ).
The specializer goes wrong for the second reason on appd (lams x : x) 0d . In this case, it attempts residualization using the rule spec app d and incorrectly produces an output program that is not completely residual
(since lams x : x is eliminable).
The property that the specializer never goes wrong (i.e., the binding-time analysis always yields consistent
directives) is a generalization of the type-soundness property for standard type systems. To establish typesoundness, one typically proves a subject-reduction result showing that typing is maintained under evaluation.
To establish consistency of directives, we prove that specialization typing is maintained under specialization.
Furthermore, we show that specialization results are always answers of the appropriate specialization type, i.e.,
that the ans is always satis ed (as promised in Section 4.2).
This ensures that the specializer will not go wrong for the rst reason. For example, in the rule spec app s,
the rule bta app s guarantees that w0 always has specialization type '1 '2 and if w0 spec a0 then a0
lams x : w00 since only appropriate answers are produced.
This also ensures that the specializer will not go wrong for the second reason. For example, in the rule
spec app d, the rule bta app d guarantees that w0 always has specialization type dyn and if w0 spec a0 then a0
must be completely residual (similarly for a1 ).
+

!

+



+

Theorem 1 (Correctness of binding-time analysis) If  w spec a and ?d bta e :  [w : ' ] and  is
compatible with ?d , then there exists e0  such that ?d bta e0 :  [a : ' ] and A ans a : ' where A is
`

2

compatible with ?d .

+

`

`

`

Proof: For notational convenience, we write :: when is a deduction of judgement and state deduction
D

J

D

J

rules in a linear format. For example, a specialization deduction that ends with the rule spec app d is written
spec app d( 0 ; 1 ) :: appd w0 w1 spec appd a0 a1 where i :: wi spec ai (i = 1; 2). The notation is somewhat
imprecise since we do not use an explicit discharge function for assumptions [19, Section 4.1]. However, the Elf
encoding makes matters suciently clear.
For the theorem hypotheses, let :: w spec a and :: bta e :  [w : ' ]. We show an e ective method
for constructing deductions :: bta e0 :  [a : ' ] and :: ans a : ' where the compatibility constraints on
undischarged assumptions are satis ed. The proof proceeds by induction on the pair of deductions and ,
S

S

+

S

S

C

+

+

B

D

S

B

Note that the order of arguments to the deduction constructors is the reverse of what one might expect since we use <- (instead
of ->) in the encodings of the binding-time and specialization logics.
6

i.e., the method is primitive recursive. Although this primitive recursive method cannot be represented in Elf
as a function (since it is not schematic), it can be represented as a relation via the following judgement.
t1 : spec W A -> bta E T W P -> bta E' T A P -> ans A P -> type.

Each case of the constructive proof is formalized as a rule for the t1 judgement. Below we show three illustrative
cases (each increasing in complexity).
case = spec z s :: 0s spec 0s and = bta z s :: bta 0 : nat [0s : sta]:
The required deductions are bta z s :: bta 0 : nat [0s : sta] and ans z s :: ans 0s : sta. This is formalized by
the following axiom.
+

S

B

t1_z_s : t1 (spec_z_s) (bta_z_s) (bta_z_s) (ans_z_s).

= spec app d( 0 ; 1 ) :: appd w0 w1 spec appd a0 a1
= bta app d( 0 ; 1 ) :: bta app e0 e1 : 2 [appd w0 w1 : dyn] :
Applying the inductive hypothesis to 0 and 0 gives deductions 0 :: bta e00 : 1 2 [a0 : dyn] and 0 ::
ans a0 : dyn. Applying the inductive hypothesis to 1 and 1 gives deductions 1 :: bta e01 : 1 2 [a1 : dyn]
and 1 :: ans a1 : dyn. The required deductions are bta app d( 0 ; 1) and ans app d( 0 ; 1 ). This is
formalized by the following rule (the arguments to the proof constructors appear in reverse order (e.g.,
spec_app_d S1 S0 instead of spec_app_d S0 S1) due to the use of <- (instead of ->) in the encodings of
binding-time and specialization logic).

case

D

D

t1_app_d : t1 (spec_app_d S1 S0) (bta_app_d B1 B0)
(bta_app_d C1 C0) (ans_app_d D1 D0)
<- t1 S0 B0 C0 D0
<- t1 S1 B1 C1 D1.

In operational terms, the inductive hypotheses are manifested as recursive calls to the function represented
by t1.
case = spec app s( 0 ; 1 ) :: apps w0 w1 spec a
= bta app s( 0 ; 1 ) :: bta app e0 e1 : 2 [apps w0 w1 : '2] :
Applying the inductive hypothesis to 0 and 0 gives deductions 00 :: bta e00 : 1 2 [a0 : '1 '2 ] and
lams y : w00 .
0 :: ans a0 : '1 '2. An examination of the ans rules (Figure 15) reveals that a0
Then, an examination of the axiomatization of the bta judgement (Figure 9) shows that we must have
00
0
00
0
0
0 = bta lam s( 0 :: bta e0 : 2 [w0 : '2 ]) :: bta lam x : e : 1 2 [lams y : w0 : '1 '2 ] where the undischarged assumptions of 0 include bta x : 1 [y : '1 ]. A simple substitution lemma (which we omit) allows
us to replace each of these assumptions in 0 with the deduction 1 :: bta e1 : 1 [w1 : '1 ] to obtain a deduction 2 :: bta e000 [x := e1 ] : 2 [w00 [y := w1] : '2 ]. Applying the inductive hypothesis to 1 and 2 gives the
required deductions :: bta e0 : 2 [a : '2] and :: ans a : '2 . This is formalized as follows.

S

B

D

t1_app_s : t1 (spec_app_s S1 S0) (bta_app_s B1 B0) C D
<- t1 S0 B0 (bta_lam_s C0) D0
<- t1 S1 (C0 _ _ B1) C D.

The \examination of the rules" (that told us deduction 00 must have bta lam s as its last rule) is captured by
matching (i.e., (bta_lam_s C0)). The required substitution lemma appears for free due to the higher-order
abstract syntax representation and the Transitivity derived rule of the LF calculus [19, Section 2.3]; it is
captured by (C0 _ _ B1). The underscores correspond to the terms e1 and w1 above. Using the underscores
and letting Elf reconstruct allows us to encode the rule more concisely.
The complete set of rules is given in Appendix A.
C

The following Elf query illustrates the function induced by the t1 rules. Given the specialization deduction
of Section 5.2 and a binding-time deduction for the term being specialized (i.e., the second annotated term
of Section 5.1), t1 constructs a binding-time deduction C for the specialization answer as well as an answer
deduction D proving that the answer is completely residual.

S

?- t1 (spec_app_s
(spec_lam_d [y2:sexp] [S:spec y2 y2] spec_lift (spec_app_s spec_z_s spec_lam_s))
spec_lam_s)
(bta_app_s bta_z_s
(bta_lam_s [x1:exp] [y1:sexp] [B1:bta x1 nat y1 sta]
bta_lam_d [x2:exp] [y2:sexp] [B2:bta x2 nat y2 dyn]
bta_lift
(bta_app_s B1
(bta_lam_s [x3:exp] [y3:sexp] [B3:bta x3 nat y3 sta] B3))))
C D.
Solving...
D = ans_lam_d [y2:sexp] [D2:ans y2 dyn] ans_z_d,
C = bta_lam_d [x2:exp] [y2:sexp] [B2:bta x2 nat y2 dyn] bta_z_d.

Elf type-checking mechanically veri es that deductions C and D are well-formed when they exist. However,
veri cation that such deductions always exist (i.e., that t1 is total) cannot be captured in Elf. This phase of
veri cation (called schema checking) must be done by hand, although its automation is the subject of current
research [32]. Our de nition makes t1 total because we give a rule for each possible pair of spec and bta rules
(giving primitive recursive structure). In addition, the deduction of answer judgements tell us that matching is
used (in rules t1 app s and t1 lift) only when it will always succeed.

6.2 Soundness of specialization

A specializer is sound if its steps describe a meaning preserving transformation on the unannotated object
program. Speci cally, if (a) e1 is the unannotated version of the program to be specialized (with the static data
already supplied), and (b) e2 is the unannotated version of the result of specialization, then a sound specializer
should guarantee that e1 and e2 are operationally equivalent (as de ned in Section 3.3).
This ensures that e1 and e2 will behave the same under evaluation when dynamic data is supplied. Specifically, for all dynamic data d such that T tp e1 [x := d] : nat and T tp e2 [x := d] : nat, e1 [x := d]
e2 [x := d]. The last claim follows since the substitution [x := d] can be represented with the context
app (lam x : [ ]) d and the fact that is easily shown to be a congruence.
We prove the operational equivalence of e1 and e2 by appealing to a program calculus for  (i.e., an equational
theory for deducing operational equivalences). Gunter [14, Chapter 4] gives a calculus for PCF, and Figure 17
recasts the calculus in our setting. e1 = e2 is the judgement expressing convertibility in the calculus. The
rules are generally the expected ones for call-by-name. However, note the use of the judgement val in the
rules conv prsc and conv ifsc. This re ects the strictness of the primitive operation pr and the strictness of the
conditional if0 in the test position. Appendix B.1 gives the Elf encoding of the calculus. It is similar to the
encoding given by Pfenning for the pure untyped -calculus [30].
The following theorem expresses the soundness of the calculus for reasoning about  operational equivalences.
`



`





Theorem 2 (Soundness of calculus) For all e1 ; e2  such that T tp e1 :  and T tp e2 :  ,
2

e1 = e2 implies e1

`



`

e2 :

Existing techniques for proving the theorem are fairly involved. For example, Gunter [14] appeals to an
adequacy result connecting the operational semantics and a model. Other techniques use co-induction principles
[12]. Unfortunately, it seems impossible to encode any of these techniques in Elf (Section 8 discusses the encoding
of these techniques in other frameworks).
The inability to encode the proof in Elf does not seem detrimental to our approach. Historically, one of
the reasons for introducing for a program calculus is to avoid repeated appeals to direct proofs of operational
equivalences and reason instead using the simple rules of the calculus. The appropriateness of this approach is
underscored by the situation we face here | it is much easier to encode a proof that appeals to the calculus.
The following theorem captures the fact that the unannotated object program e is operationally equivalent
to the unannotated specialization result e0 .

conv beta : app (lam x : e0) e1 = e0 [x := e1 ]
conv x :

x x: e = e[x := x x: e]

conv prz :

pr 0 = 0

conv prsc :

val e
pr (sc e) = e

conv ifz :

if0 0 e2 e3 = e2

conv ifsc :

val e
if0 (sc e) e2 e3 = e3

conv re :

e = e

conv sym :

e 2 = e 1
e1 = e2

conv trans :

e1 = e2 e2 = e3
e1 = e3

conv sc c :

e = e0
sc e = sc e0

e = e0
pr e = pr e0

conv pr c :
conv if c1 :
conv if c2 :
conv if c3 :

e1 = e01
if0 e1 e2 e3 = if0 e01 e2 e3
e2 = e02
if0 e1 e2 e3 = if0 e1 e02 e3
e3 = e03
if0 e1 e2 e3 = if0 e1 e2 e03

conv lam c :

e = e0
lam x : e = lam x : e0

conv app c0 :

e0 = e00
app e0 e1 = app e0 e1

conv app c1 :

e1 = e01
app e0 e01 = app e0 e01

Figure 17: Program calculus for 

Theorem 3 (Soundness of specializer) If  ` w +spec a and ?d ` bta e :  [w : ' ] and  is compatible
with ?d , then there exists e0 2  such that ?d ` bta e0 :  [a : ' ] and e = e0 .
Proof: To formalize the proof, we construct a function (via a relation as in Theorem 1) which builds a deduction
showing e converts to e0 .
t3 : spec W A -> bta E T W P -> bta E' T A P -> ans A P -> conv E E' -> type.

The t3 judgement expresses a function which takes two proofs as input: a proof of spec W A and a proof of
bta E T W P. The output judgement bta E' T A P acts as the unannotating function from bt terms A to 
terms E' (see Section 4.1.3).
The output proof of ans A P is necessary for extracting value deductions to be used in the conv prsc and
conv ifsc rules of the calculus. This requires a small lemma | essentially stating that if the specialization
type P is static, then dropping the annotations on the answer A yields a value. In other words, given a proof
bta E T W P, a proof ans W P that W is a value, and a proof that P is static, we can construct a proof that the
de-annotation of W (given by E) is a value. This is formalized with the following judgement.
ans_to_val : bta E T W P -> ans W P -> static P -> val E -> type.

This proof is simple and omitted. The Elf encoding is given in Appendix B.2 along with the encoding of
predicates describing static and dynamic specialization types.
Finally, the conv E E' judgement expresses the desired property that E converts to E'. A strategy similar
to the one used in the proof of Theorem 1 gives the rules de ning t3 (see Appendix B.3).
Note that by including bta E' T A P and ans A P as we generate the deduction of conv E E', we are in
essence reproving Theorem 1. An alternative approach would be to use Theorem 1 to justify some of the steps
in the proof below. This would be manifested in the Elf encoding as \calls" to the goal represented by the t1
judgement. We choose the formalization below to simplify the presentation.

7 Related work
Despeyroux rst emphasized using deductive systems to de ne transformations [8]. She speci ed a compiler, and
source and target language semantics using deductive systems. The speci cations were executed via encodings
into Typol. Informal proofs of correctness were given as relations between deductions. However, these proofs
could not be formalized in Typol because it does not support the direct manipulation of its own deductions.
Hannan and Pfenning [18] improved upon this by formalizing similar proofs of correctness in Elf. Elf (unlike
e.g., Typol, and -Prolog) supports the direct manipulation of its own deductions. Furthermore, Elf type
checking mechanically veri es that deductions corresponding to correctness proofs are well-formed. Other work
using Elf to encode operational semantics includes Michaylov and Pfenning's encoding of call-by-value natural
semantics and associated meta-theory for the language Mini-ML [23], Pfenning's representation of a ChurchRosser proof [30], Hannan's encoding of a type system for closure conversion [16], and Niss and the author's
representation of the correctness proofs for the familiar translation of call-by-name to call-by-value using thunks
[26].
Despeyroux [8, Section 8], and Hannan and Pfenning [18, Section 7], suggested that their methods could be
used to specify \mixed computation" and partial evaluation, respectively. Hannan and Miller [17] carried out
Despeyroux's suggestion; they use deductive systems encoded in -Prolog to obtain executable speci cations of
mixed computation (in their work, \mixed computation" = non-deterministic on-line partial evaluation).
Our contributions include using deductive systems to specify program specialization directed by type-based
analysis. Moreover, we adapt the techniques of Hannan and Pfenning [18], and Michaylov and Pfenning [23] to
mechanically verify correctness.
In doing so, we obtain veri ed proofs similar in scope to the ones given by Gomard and Jones [10, 11] for
-mix. Their denotational meta-language is signi cantly more complex than our logic-based meta-language and
it is unlikely that the proofs there could be formalized or mechanically veri ed to the extent that we have done
here. However, it must be noted that one of their goals was self-application. Relying on the similarity between
their meta-language and object-language, they obtain an object-language speci cation of the partial evaluator

(giving self-applicability) by a fairly easy (though informal and unveri ed) translation from the meta-language
speci cation. In our setting this is more dicult since the character of our meta-language (logical) is quite
di erent from our object language (functional).
Building upon the work of Gomard, Jones, and Mogensen, Palsberg [27] and Wand [37] give detailed presentations of the correctness of binding-time analysis. Palsberg presents a generalization of Gomard and Jones
criteria [11] for consistent binding-time annotations. Wand studies Mogensen's self-applicable partial evaluator
[24] for the pure -calculus. His binding-time analysis is essentially the same as Gomard and Jones's, as well
as the one presented here. Wand's goals with respect to correctness of the analysis and specializer are more
ambitious than those here, because his correctness criteria is strong enough to specify the behaviour of the
partial evaluator when self-applied.
It would be interesting to see to what extent the meta-theory used by Palsberg and Wand could be formalized
in Elf. In particular, once Wand has established the correctness results for an interpreter and partial evaluator,
his proofs for the Futamura projections have a simple calculational style (based on equational reasoning). The
proofs follow the same steps as the \Mix equations" [20, Chapter 1] which are often used to reason about the
results of self-application. We are currently investigating how these reasoning principles can be formalized (e.g.,
to obtain a \Mix equational theory") and encoded in LF.
In recent work, Davies and Pfenning [7] give a type system for expressing staged computation based on
the intuitionistic modal logic S4. They have implemented the type system and a portion of the associated
correctness proofs in Elf.
Our work focuses on partial evaluation of functional programs, but similar techniques can be applied to
imperative languages as well. Blazy and Facon [2] specify a simple specializer for FORTRAN using natural
semantics, and derive a prototype from the speci cation using the Centaur programming environment. However,
their correctness proofs are not formalized since the Centaur environment is not oriented toward this purpose.
Bertot and Fraer [1] show how similar correctness proofs (for a specializer for an imperative language) can be
formalized and mechanically checked using Coq.

8 Conclusion
8.1 Summary

We have speci ed the main components of a simple oine partial evaluator (i.e., binding-time constraints and
a program specializer) using natural-deduction style logics. These speci cations were formalized by encoding
them in LF. Prototypes were obtained directly from the formal speci cations using Elf. We formalized and
mechanically veri ed a signi cant portion of the meta-theory of oine partial evaluation (e.g., correctness of
binding-time analysis and soundness of the specializer) via meta-programming in Elf. A certain degree of
synergy is obtained by using the declarative formalism of deductive systems: one may specify, prototype and
mechanically verify correctness via meta-programming | all within a single framework.

8.2 Assessment
Using deductive systems to specify analyses and transformations: The simplicity of the speci cations

stems from their declarative nature (which allows inessential algorithmic details and model theory to be avoided).
This does not mean that other formalisms such as denotational semantics are not useful. There are certainly
situations where the extra reasoning capabilities o ered by model theory are bene cial. However, it seems that
in many applications | particularly in the area of program specialization | these extra capabilities are never
utilized and other styles of formalization such as the ones used here may be more appropriate.

Using LF as a meta-language for de ning speci cations: Using a formal meta-language such as LF

forces one to be precise, and such precision usually lends clarity to presentations. However, the rigidness of
LF almost demands that one cast speci cations as natural-deduction-style logics and use higher-order abstract
syntax. This style of logic may not be appropriate in all situations. In addition, it would be interesting to see
how our speci cations and proofs would be formalized in other logical frameworks. Recent work suggests that it
may be possible to generate program analyses from formal speci cations. LF stands as a possible meta-language
for formalizing such speci cations.

Using Elf to obtain executable speci cations: We found Elf particularly useful for prototyping the

binding-time analysis. It should be investigated if this scales up to more complicated type-based analyses that
use conjunctive types and more general forms of subtyping. More complex speci cations may well be prone
to looping due to the simplistic depth- rst search strategy used by Elf. However, we have successfully used
the techniques presented here to prototype a more complex higher-order version of the multi-level binding-time
analysis of Gluck and Jrgensen [9].

Using Elf to mechanically verify correctness: It seems possible to use these techniques to verify other
forms of program specialization. For example, if one takes the view that an annotated program is its own
generating extension, the techniques here are very close to what one would use to verify the correctness of a
hand-written cogen [20]. It remains to be seen if these techniques scale up to more robust forms of partial
evaluation that include sophisticated folding strategies. One approach might be to use Sand's calculus for sound
fold/unfold transformations [34].
The logic of Elf was not strong enough to encode the calculus soundness proof (see Theorem 2). More
generally, the induction principles as required in the proofs of Theorems 1 and 3 could not be completely
formalized. This manifested itself in the inabilty to mechanically guarantee the totality of the functions (de ned
as relations) representing the constructive proofs. Frameworks based on stronger logics (such as Isabelle-HOL
and Isabelle-ZF [28], HOL [13], and Coq) can overcome these problems. Nipkow [25] has used Isabelle-HOL to
encode a large portion of the theory given in Winskel's semantics textbook [39]. Similar techniques would be
used in formalizingthe calculus-soundness proof using model theory. Collins [4] has used HOL to encode a theory
of bi-similarity for higher-order functional language. This could easily be extended to prove the soundness of the
calculus using coinduction techniques. Finally, the forthcoming thesis of Welinder [38] contains a formalization
of various principles required for reasoning about specialization of rst-order functional programs.
Using stronger logics leads to tradeo s. Proofs become much more involved (as formalization is more
complete). Prototypes are harder to obtain directly from speci cations.

A Correctness of binding-time analysis
%%% definition of t1 judgement
t1 : spec W A -> bta E T W P -> bta E' T A P -> ans A P -> type.
%%% definition of t1 axiom and rules
% static constructs
t1_z_s : t1 (spec_z_s) (bta_z_s) (bta_z_s) (ans_z_s).
t1_sc_s : t1 (spec_sc_s S) (bta_sc_s B) (bta_sc_s C) (ans_sc_s D)
<- t1 S B C D.
t1_prz_s : t1 (spec_prz_s S) (bta_pr_s B) (bta_z_s) (ans_z_s)
<- t1 S B (bta_z_s) D.
t1_prsc_s : t1 (spec_prsc_s S) (bta_pr_s B) C D
<- t1 S B (bta_sc_s C) (ans_sc_s D).
t1_ifz_s : t1 (spec_ifz_s S2 S1) (bta_if_s B3 B2 B1) C2 D2
<- t1 S1 B1 (bta_z_s) D1
<- t1 S2 B2 C2 D2.
t1_ifsc_s : t1 (spec_ifsc_s S3 S1) (bta_if_s B3 B2 B1) C3 D3
<- t1 S1 B1 (bta_sc_s C1) D1
<- t1 S3 B3 C3 D3.
t1_lam_s : t1 (spec_lam_s) (bta_lam_s B) (bta_lam_s B) (ans_lam_s).
t1_app_s : t1 (spec_app_s S1 S0) (bta_app_s B1 B0) C D
<- t1 S0 B0 (bta_lam_s C0) D0
<- t1 S1 (C0 _ _ B1) C D.
t1_fix_s : t1 (spec_fix_s S) (bta_fix_s B) C D
<- t1 S (B (fix E) (bfix s W) (bta_fix_s B)) C D.
% dynamic constructs

t1_z_d : t1 (spec_z_d) (bta_z_d) (bta_z_d) (ans_z_d).
t1_sc_d : t1 (spec_sc_d S) (bta_sc_d B) (bta_sc_d C) (ans_sc_d D)
<- t1 S B C D.
t1_pr_d : t1 (spec_pr_d S) (bta_pr_d B) (bta_pr_d C) (ans_pr_d D)
<- t1 S B C D.
t1_if_d : t1 (spec_if_d S3 S2 S1) (bta_if_d B3 B2 B1)
(bta_if_d C3 C2 C1) (ans_if_d D3 D2 D1)
<- t1 S1 B1 C1 D1
<- t1 S2 B2 C2 D2
<- t1 S3 B3 C3 D3.
t1_lam_d : t1 (spec_lam_d S) (bta_lam_d B) (bta_lam_d C) (ans_lam_d D)
<- {x} {y} {S': spec y y} {B' : bta x T1 y dyn} {D': ans y dyn}
t1 S' B' B' D'
-> t1 (S y S') (B x y B') (C x y B') (D y D').
t1_app_d : t1 (spec_app_d S1 S0) (bta_app_d B1 B0)
(bta_app_d C1 C0) (ans_app_d D1 D0)
<- t1 S0 B0 C0 D0
<- t1 S1 B1 C1 D1.
t1_fix_d : t1 (spec_fix_d S) (bta_fix_d B) (bta_fix_d C) (ans_fix_d D)
<- {x} {y} {S': spec y y}
{B' : bta x (arrow T1 T2) y dyn} {D': ans y dyn}
t1 S' B' B' D'
-> t1 (S y S') (B x y B') (C x y B') (D y D').
t1_lift : t1 (spec_lift S) (bta_lift B) (bta_z_d) (ans_z_d)
<- t1 S B (bta_z_s) D.
t1_liftsc : t1 (spec_liftsc S2 S1) (bta_lift B1) (bta_sc_d C2) (ans_sc_d D2)
<- t1 S1 B1 (bta_sc_s C1) D1
<- t1 S2 (bta_lift C1) C2 D2.

B Soundness of specializer

B.1 Convertability relation
%%% convertability judgement
conv : exp -> exp -> type.

%%% convertability axioms and rules
% axioms
conv_beta : conv (app (lam E0) E1) (E0 E1).
conv_fix : conv (fix E) (E (fix E)).
conv_prz : conv (pr z) z.
conv_prsc : conv (pr (sc E)) E
<- val E.

conv_ifz : conv (if z E2 E3) E2.
conv_ifsc : conv (if (sc E1) E2 E3) E3
<- val E1.
% inference rules
conv_refl : conv E E.
conv_sym : conv E1 E2
<- conv E2 E1.
conv_trans : conv E1 E3
<- conv E1 E2
<- conv E2 E3.
% context rules
conv_sc_c: conv (sc E) (sc E')
<- conv E E'.
conv_pr_c: conv (pr E) (pr E')
<- conv E E'.
conv_if_c1: conv (if E1 E2 E3) (if E1' E2 E3)
<- conv E1 E1'.
conv_if_c2: conv (if E1 E2 E3) (if E1 E2' E3)
<- conv E2 E2'.
conv_if_c3: conv (if E1 E2 E3) (if E1 E2 E3')
<- conv E3 E3'.
conv_lam_c: conv (lam E1) (lam E2)
<- {x} (conv (E1 x) (E2 x)).
conv_fix_c: conv (fix E1) (fix E2)
<- {x} (conv (E1 x) (E2 x)).
conv_app_c0: conv (app E0 E1) (app E0' E1)
<- conv E0 E0'.
conv_app_c1: conv (app E0 E1) (app E0 E1')
<- conv E1 E1'.

B.2 Extracting value deductions from answer deductions
B.2.1 Predicates classifying specialization types
%%% judgements
static : styp T -> type.
dynamic : styp T -> type.
%%% static rules
static_nat : static sta.
static_arrow : static (barrow P1 Pw).

%%% dynamic rules
dynamic_dyn : dynamic dyn.

B.2.2 Function from static answers to values
%%% ans_to_val judgement
ans_to_val : bta E T W P -> ans W P -> static P -> val E -> type.
%%% ans_to_val rules
ans_to_val_z_s : ans_to_val bta_z_s ans_z_s static_nat val_z.
ans_to_val_sc_s : ans_to_val (bta_sc_s B) (ans_sc_s A) static_nat (val_sc V)
<- ans_to_val B A S V.
ans_to_val_lam_s : ans_to_val (bta_lam_s B) ans_lam_s static_arrow val_lam.

B.3 Proof of Theorem 3
%%% definition of t3 judgement
t3 : spec W A -> bta E T W P -> bta E' T A P -> ans A P -> conv E E' -> type.
%%% definition of t3 axiom and rules
% static constructs
t3_z_s : t3 (spec_z_s) (bta_z_s) (bta_z_s) (ans_z_s) (conv_refl).
t3_sc_s : t3 (spec_sc_s S) (bta_sc_s B) (bta_sc_s C) (ans_sc_s D) (conv_sc_c E)
<- t3 S B C D E.
t3_prz_s : t3 (spec_prz_s S) (bta_pr_s B) (bta_z_s) (ans_z_s)
(conv_trans (conv_prz) (conv_pr_c E))
<- t3 S B (bta_z_s) D E.
t3_prsc_s : t3 (spec_prsc_s S) (bta_pr_s B) C D
(conv_trans (conv_prsc V)
(conv_pr_c E))
<- ans_to_val C D static_nat V
<- t3 S B (bta_sc_s C) (ans_sc_s D) E.
t3_ifz_s : t3 (spec_ifz_s
(conv_trans
<- t3 S1 B1
<- t3 S2 B2

S2 S1) (bta_if_s B3 B2 B1) C2 D2
E2 (conv_trans (conv_ifz) (conv_if_c1 E1)))
(bta_z_s) D1 E1
C2 D2 E2.

t3_ifsc_s : t3 (spec_ifsc_s S3 S1) (bta_if_s B3 B2 B1) C3 D3
(conv_trans E3 (conv_trans (conv_ifsc V) (conv_if_c1 E1)))
<- ans_to_val C1 D1 static_nat V
<- t3 S1 B1 (bta_sc_s C1) (ans_sc_s D1) E1
<- t3 S3 B3 C3 D3 E3.
t3_lam_s : t3 (spec_lam_s) (bta_lam_s B) (bta_lam_s B) (ans_lam_s) (conv_refl).
t3_app_s : t3 (spec_app_s S1 S0) (bta_app_s B1 B0) C D

(conv_trans E1 (conv_trans (conv_beta) (conv_app_c0 E0)))
<- t3 S0 B0 (bta_lam_s C0) D0 E0
<- t3 S1 (C0 _ _ B1) C D E1.
t3_fix_s : t3 (spec_fix_s S) (bta_fix_s B) C D
(conv_trans E (conv_fix))
<- t3 S (B (fix E0) (bfix s W0) (bta_fix_s B)) C D E.
% dynamic constructs
t3_z_d : t3 (spec_z_d) (bta_z_d) (bta_z_d) (ans_z_d) (conv_refl).
t3_sc_d : t3 (spec_sc_d S) (bta_sc_d B) (bta_sc_d C) (ans_sc_d D) (conv_sc_c E)
<- t3 S B C D E.
t3_pr_d : t3 (spec_pr_d S) (bta_pr_d B) (bta_pr_d C) (ans_pr_d D) (conv_pr_c E)
<- t3 S B C D E.
t3_if_d : t3 (spec_if_d S3 S2 S1) (bta_if_d B3 B2 B1)
(bta_if_d C3 C2 C1) (ans_if_d D3 D2 D1)
(conv_trans (conv_if_c3 E3)
(conv_trans (conv_if_c2 E2)
(conv_if_c1 E1)))
<- t3 S1 B1 C1 D1 E1
<- t3 S2 B2 C2 D2 E2
<- t3 S3 B3 C3 D3 E3.
t3_lam_d : t3 (spec_lam_d S) (bta_lam_d B) (bta_lam_d C) (ans_lam_d D)
(conv_lam_c E)
<- {x} {y} {S': spec y y} {B' : bta x T3 y dyn} {D': ans y dyn}
{E': conv x x}
t3 S' B' B' D' E'
-> t3 (S y S') (B x y B') (C x y B') (D y D') (E x).
t3_app_d : t3 (spec_app_d S1 S0) (bta_app_d B1 B0)
(bta_app_d C1 C0) (ans_app_d D1 D0)
(conv_trans (conv_app_c1 E1) (conv_app_c0 E0))
<- t3 S0 B0 C0 D0 E0
<- t3 S1 B1 C1 D1 E1.
t3_fix_d : t3 (spec_fix_d S) (bta_fix_d B) (bta_fix_d C) (ans_fix_d D)
(conv_fix_c E)
<- {x} {y} {S': spec y y}
{B' : bta x (arrow T1 T3) y dyn} {D': ans y dyn} {E': conv x x}
t3 S' B' B' D' E'
-> t3 (S y S') (B x y B') (C x y B') (D y D') (E x).
t3_lift : t3 (spec_lift S) (bta_lift B) (bta_z_d) (ans_z_d) E
<- t3 S B (bta_z_s) D E.
t3_liftsc : t3 (spec_liftsc S2 S1) (bta_lift B1) (bta_sc_d C2) (ans_sc_d D2)
(conv_trans (conv_sc_c E2) E1)
<- t3 S1 B1 (bta_sc_s C1) D1 E1
<- t3 S2 (bta_lift C1) C2 D2 E2.

